A Second One-Compartment Problem

```
> restart;
```

The IVP below models the following situation: A 200 gallon tank that is always thoroughly mixed originally contains 50 gal of salt water with a salt concentration of .1 lb/gal. A salt water solution with a salt concentration of .25 lb/gal is pumped into the tank at a rate of 6 gal/min while at the same time salt water is flowing out of the tank at 4 gal/min. We want to find a formula giving the concentration of salt in the tank up to the point of overflowing. Since the volume of the tank is increasing by 2 gal/min, this will take 75 minutes.

We let $x(t)$ represent the amount of salt (in lbs) in the tank, $V(t)$ represent the volume (in gal) of the tank, and $c(t)$ represent the concentration (in lbs/gal) of salt in the tank, all at time t. We are searching for a formula for $c(t) = \frac{x(t)}{V(t)}$. We first need a formula for V(t) as a function of t because the outflow rate of salt will have to be expressed in terms of it.

```
> V:=t->50+2*t;
```

$$V := t \to 50 + 2\,t$$

We next use an IVP to model the rate of change of the amount of salt in the tank.

```
> deq:=diff(x(t),t)=6*.25-4*x(t)/V(t);
```

$$deq := \frac{\mathrm{d}}{\mathrm{d}t} x(t) = 1.50 - \frac{4\,x(t)}{50 + 2\,t}$$

```
> IC:=x(0)=5;
```

$$IC := x(0) = 5$$

We solve the IVP for $s(t)$.

```
> soln:=dsolve({deq,IC},x(t));
```

$$soln := x(t) = \frac{25}{2} + \frac{1}{2}\,t - \frac{9375}{2\,(25 + t)^2}$$

We simplify the right hand side of this solution.

```
> x:=simplify(rhs(soln));
```

$$x := \frac{1}{2}\,\frac{t^3 + 75\,t^2 + 1875\,t + 6250}{(25 + t)^2}$$

We use unapply to change s to a function of t.

```
> x:=unapply(x,t);
```

$$x := t \to \frac{1}{2}\,\frac{t^3 + 75\,t^2 + 1875\,t + 6250}{(25 + t)^2}$$

We can now find the formula for the function $c(t)$ as a quotient.

```
> x:=unapply(simplify(x(t)/V(t)),t);
```

$$x := t \to \frac{1}{4}\,\frac{t^3 + 75\,t^2 + 1875\,t + 6250}{(25 + t)^3}$$

Finally, we plot the graph of the solution over our 75 minute time frame.

```
> plot(x(t),t=0..75,color=blue,labels=["t","c(t)"],title=
  "Concentration of Salt in a Tank",titlefont=[TIMES,ROMAN,12]);
```



Is this the graph you expected?